
	Reason for change:

This needs to be corrected.

Also, the current TTCN implementation of the while loop does not execute for another iteration when PDU Session Modification Complete is expected. This also needs to be accommodated.  

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc83237155]Corrections required
0.1 [bookmark: _Toc83237156]Correction to function f_NR5GC_IRAT_S1ToN1_SessionModification
	Function name
	f_NR5GC_IRAT_S1ToN1_SessionModification

	Reason for change
	After implementation of R5-215678, this function was modified to be able to hande the rrcReconfigurationComplete and PDU Session Modification Complete coming in any order.

However, the TTCN implementation needs tbe corrected further. In current TTCN implementation, depending on variable v_WaitForRRCComplete the TTCN expects a rrcReconfigurationComplete. But for this, the SRB is hardcoded to SRB2, and the template used is cr_NG_NAS_IndWithPiggybacking . This is incorrect as the rrcReconfigurationComplete will come on SRB1, not SRB2. Also the teemplate used to capture this should be cr_38508_RRCReconfigurationComplete. 
This needs to be corrected.

Also, the current TTCN implementation of the while loop does not execute for another iteration when PDU Session Modification Complete is expected. This also needs to be accommodated.  

	Summary of change
	Updated the TTCN to handle the rrcReconfigurationComplete on SRB1. Also introduced a new variable v_WaitForPDUSessionModificationComplete and based on this value extended the while loop to handle the PDU Session Modification Complete.
Please see below.

	TTCN module
	[bookmark: _GoBack]NR5GC_IRAT_CommonFunctions.ttcn

	MCC160 Comment
	



Before Change:
	function f_NR5GC_IRAT_S1ToN1_Steps14(NR_CellId_Type p_NR_CellId,
                                        RRCConnectionReleaseRequired_Type p_ReleaseConnection := rrcConnectionRelease,
                                        boolean p_FirstTimeS1ToN1 := true,
                                        InterworkWithoutN26 p_IwkN26 := NOT_SUPPORTED)  runs on NR5GC_PTC return NR_RadioBearerId_Type
  {
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
    //@sic R5s201060 sic@
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;
    var boolean v_FailWhenK_LessN := false;
    var template NG_Request_Type v_RequestType := cr_NG_Request_Type('001'B);
    var boolean v_MsgRxd := false;
    var integer i;
    var default v_MyDefaultVar := null;
    
    timer  t_Wait := 8.0; // @sic R5-204399, R5s210077 sic@

    if (p_IwkN26 == SUPPORTED) { // @sic R5-204401 sic@
      v_FailWhenK_LessN := true; // The only difference between 14a1 and 14b1 branch
      v_RequestType := cr_NG_Request_Type('010'B); // @sic R5-206290 sic@
    }
    
    // always wait for PDU info, even if not 1st time
    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);

    // @sic R5s210520 sic@
    // Clear store of msgs received in parallel
    f_NR5GC_MsgInDefault_ResetStoreOfPDUSessionEstReq();
    // Then activate store for any new msgs received - this works for Modification Reqs too
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(p_NR_CellId));

    t_Wait.start;  //  @sic R5-204399 sic@

    // make sure all the PDNs configured on EUTRA are changed in routing table @sic R5s210149, R5s210520 sic@
    for (i := 0; i < lengthof (v_PduSessionInfoList); i := i + 1) {
      f_IP_ChangeDrbMappingNR(IP, p_NR_CellId, cs_IP_QosFlowId(v_PduSessionInfoList[i].PDU_SessionId, v_PduSessionInfoList[i].QFI), v_PduSessionInfoList[i].PdnIndex);
    }
    // @sic R5s210520 sic@
    // has a NAS msg already been received? if so, stop waiting
    if (f_NR5GC_MsgInDefault_GetNumOfPDUSessionEstReq() > 0) {
      v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOnePDUSessionEstReq(1);
      v_MsgRxd := true;
      t_Wait.stop;
    } else {
      alt { //  @sic R5-204401 sic@
        [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp  // @sic R5s201413 sic@
          {
           f_NR5GC_MsgInDefault_StorePDUSessionEstReq (v_ReceivedAsp); // @sic R5s210687 sic@
           v_MsgRxd := true;
           t_Wait.stop;
          }
        [] t_Wait.timeout {}
      }
    }
    if (v_MsgRxd and ischosen(v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg.pdu_Session_Modification_Request)) { // @sic R5s210520 sic@
      if (p_FirstTimeS1ToN1 and p_IwkN26 == NOT_SUPPORTED) {
        f_NR5GC_IRAT_S1ToN1_SessionModification(p_NR_CellId, v_PduSessionInfoList, false, v_SRB); // @sic R5s201413 sic@
        v_SRB := tsc_NR_RbId_SRB2; // @sic R5s210687 sic@
      } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Not expecting PDU Session Modification message");
      }
      // reset flag as 1st msg now been processed
      v_MsgRxd := false;
      t_Wait.start; // restart timer in case there are now PDU Establishment messages
      alt { //  @sic R5-204401 sic@
        [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp  // @sic R5s201413 sic@
          {
          // update v_ReceivedAsp and msg store if received again  @sic R5s210520 sic@
            f_NR5GC_MsgInDefault_StorePDUSessionEstReq (v_ReceivedAsp);
            v_MsgRxd := true;
            t_Wait.stop;
          }
        [] t_Wait.timeout {
              // Do nothing as no (more) PDU Modification/Establishment messages received so carry on
          }
      }
    }
    // don't need this anymore so deactivate it before it needs reactivating
    deactivate(v_MyDefaultVar);  // @sic R5s210520 sic@

    if (v_MsgRxd) {  // @sic R5s210672 sic@
      if (ischosen(v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg.pdu_Session_Establishment_Request)) { // @sic R5s210520 sic@
        f_NR5GC_PDUSessionEstablishment (p_NR_CellId, pc_noOf_PDUsSameConnection, tsc_NR_RbId_SRB1, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, v_RequestType, -, -, -, -, -, -, "", v_FailWhenK_LessN, -, v_PduSessionInfoList, -, 1); // @sic R5-206290, R5s210149, R5s210520 sic@
        v_SRB := tsc_NR_RbId_SRB2; // @sic R5-204401 sic@
        //  @sic R5-204401 sic@
        if (pc_noOf_PDUsNewConnection >0 ){
          f_NR_RRCRelease(p_NR_CellId);
          f_NR5GC_RRC_Idle_Extension(p_NR_CellId, TESTMode_OFF, noRrcConnectionRelease);
        }
      } else {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Wrong NAS Message Received");
      }
    }

    // in case there's no PDU messages at all, don't release the connection in the above functions, but do it here instead
    if (p_ReleaseConnection == rrcConnectionRelease) {
      // Release RRC to enter RRC_IDLE
      f_NR_RRCRelease(p_NR_CellId);
    }
    return v_SRB; // @sic R5-204401 sic@
  }



After Change:
	function f_NR5GC_IRAT_S1ToN1_SessionModification(NR_CellId_Type p_NR_CellId,
                                                   EUTRA_NR_PduSessionInfoList_Type p_PduSessionsFromLTE,
                                                   boolean p_FailWhenK_LessN := true,
                                                   NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                                                   integer p_ExpectedNumberOfNewPDUSessions := (pc_noOf_PDUsSameConnection + pc_noOf_PDUsNewConnection),
                                                   boolean p_DRBsAlreadyConfigured := false) runs on NR5GC_PTC return boolean
  {
    var template (value) PDUSessionInfoList_Type v_FromLTEList;
    var NG_NAS_UL_PduList_Type v_PduSessionModificationRequestList := {};
    var IntegerList_Type v_PDU_SessionIdList := {};
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_NG_NAS_UL_Pdu;
    var NG_NAS_UL_Message_Type v_GSMMsg;
    var integer v_PDU_SessionId;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template(value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)
    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;
    var boolean v_ReregisterIMS := false;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var integer N := p_ExpectedNumberOfNewPDUSessions;  // number of PDU SESSION MODIFICATION REQUEST messages being expected
    var integer K := 0;   // number of PDU SESSION MODIFICATION REQUEST messages already processed (i.e. the corresponding PDU SESSION MODIFICATION COMMAND and PDU SESSION MODIFICATION COMPLETE have been sent/received)
    var integer L := 0;   // number of PDU SESSION MODIFICATION REQUEST messages being received
    var boolean v_WaitForComplete := false;
    var template (present) NG_UE_SM_Cap v_NG_UE_SM_Cap := ?;                         // according to 38.508-1 Table 4.5A.2C.2.3-1
    var template (present) IntegrityProtMaxDataRate v_IntegrityProtMaxDataRate := ?; // according to 38.508-1 Table 4.5A.2C.2.3-1
    var integer i;
    var SRBs_AlreadyConfigured v_SRBs := SRB1;
    var NR_RadioBearerId_Type v_SRB := p_SRB;  // @sic R5s210672 sic@
    var IntegerList_Type v_TransferredSessions;
    var PDU_PDN_DNN_Type v_PDU_Type := Internet_DNN; // assume this for now @sic R5-206296 sic@
    var boolean v_RxdMsg := false;
    var boolean v_WaitForRRCComplete := false;
    var boolean v_WaitForPDUSessionModificationComplete := false; //WA#6_2_2_2
    
    timer t_Wait := 10.0; // @sic R5-204330 sic@

    // Build the GSMMobilityInfo for all received PDUs, but don't store it (some may be established, rather than modified)
    if (ispresent(v_GMM_MobilityInfo.SessionStatus)) { // @sic R5-206290 sic@
      v_TransferredSessions := f_NG_ConvertPDU_SessionStatus(valueof(v_GMM_MobilityInfo.SessionStatus));
      for (i := 0; i < lengthof (p_PduSessionsFromLTE); i := i + 1) {
        if (f_IntegerList_Search(v_TransferredSessions, p_PduSessionsFromLTE[i].PDU_SessionId) != -1) {
          // don't store if not in transferred list
          if (p_PduSessionsFromLTE[i].IsIMS) {  // @sic R5-206296 sic@
            v_PDU_Type := IMS_DNN;
          }
          v_FromLTEList[i] := cs_GSM_MobilityInfo(int2oct(p_PduSessionsFromLTE[i].PDU_SessionId, 1), -, v_PDU_Type); // @sic R5s200564 sic@
        }
      }
    }
        
    t_Wait.start;  // @sic R5-204330 sic@
    while ((K < N) or v_WaitForRRCComplete or v_WaitForPDUSessionModificationComplete /*WA#6_2_2_2*/) { // @sic R5-215678 sic@
      if (f_NR5GC_MsgInDefault_GetNumOfPDUSessionEstReq() > L) { // @sic R5s210520 sic@
        // 1st msg was received before entering function
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOnePDUSessionEstReq(1);
        v_RxdMsg := true;
        t_Wait.stop;
      } else {
        alt { // @sic R5-204330 sic@
          //WA#6_2_2_2
          [v_WaitForRRCComplete]SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete))
            {
                v_WaitForRRCComplete := false;              
            }
          /* COMMENTED OUT [v_WaitForRRCComplete] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB2,
                                      cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp
          {
            v_WaitForRRCComplete := false;
          }*/
          [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp // @sic R5s201413, R5s210672 sic@
          {
            v_RxdMsg := true;
            t_Wait.stop;
          }
          [] t_Wait.timeout {
            if (p_FailWhenK_LessN and (K < N)) {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Not enough PDU Session Modification Request messages");
            } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop  @sic R5-204330 sic@
              break;
            }
          }
        }
      }
      if (v_RxdMsg) { // @sic R5s210520 sic@
        v_NG_NAS_UL_Pdu := v_ReceivedAsp.Signalling.Nas[0].Pdu;
        v_GSMMsg := v_NG_NAS_UL_Pdu.PiggybackedPduList[0].Msg;

        if (ischosen(v_GSMMsg.pdu_Session_Modification_Request)) {
          v_PduSessionModificationRequestList[L] := v_NG_NAS_UL_Pdu;
          L := L + 1;
          if (L > N) { // @sic R5-204330 sic@
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "L > N");
          }
          if (not v_WaitForComplete and (K < L)) {   // => process the next PDU SESSION MODIFICATION REQUEST
            v_NG_NAS_UL_Pdu := v_PduSessionModificationRequestList[K];

            if (not f_Check_NG_PDUSessionModificationReq(v_GSM_MobilityInfo, valueof(v_FromLTEList), v_NG_NAS_UL_Pdu, -, -, v_NG_UE_SM_Cap, -, -, -, v_IntegrityProtMaxDataRate, -, -, -, -, -, cr_NG_Request_Type('101'B))) {     // v_GSM_MobilityInfo is out parameter of f_Check_NG_PDUSessionModificationReq
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION REQUEST");
            }
            // ok, this PDU is to be modified, so can now store it in our list
            v_PDU_SessionId := oct2int(v_GSM_MobilityInfo.SessionId); // @sic R5s200564 sic@
            if (f_IntegerList_Search(v_PDU_SessionIdList, v_PDU_SessionId) < 0) {
              v_PDU_SessionIdList[K] := v_PDU_SessionId;
            } else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "There has been a PDU SESSION MODIFICATION REQUEST with the same session id already");
            }
            for (i := 0; i < N; i := i + 1) {
              if (p_PduSessionsFromLTE[i].PDU_SessionId == oct2int(v_GSM_MobilityInfo.SessionId)) {
     //             f_IP_ChangeDrbMappingNR(IP, p_NR_CellId, cs_IP_QosFlowId(p_PduSessionsFromLTE[i].PDU_SessionId, p_PduSessionsFromLTE[i].QFI), p_PduSessionsFromLTE[i].PdnIndex);
                v_GSM_MobilityInfo.EPS_Bearer := valueof(p_PduSessionsFromLTE[i].EPS_Bearer); // @sic R5s210408 sic@
                v_GSM_MobilityInfo.PdnIndex := valueof(p_PduSessionsFromLTE[i].PdnIndex); // @sic R5s210408 sic@
                v_ReregisterIMS := v_ReregisterIMS or p_PduSessionsFromLTE[i].IsIMS;
                break;
              }
            }
            f_NR5GC_MobileInfo_AddPDUSession(v_GSM_MobilityInfo); // @sic R5s201413, R5s210408 sic@
            // At the moment no DRBs are configured in the prose - so this will only work for HO TCs
            if (v_ReregisterIMS) { // @sic R5-205942 sic@
              // Tell IMS to start waiting for the Register message.
              f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_ReregistrationStart(NR));
            }
            v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo);
            f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,
                                             v_CellGroupConfig,
                                             v_RadioBearerList,
                                             v_SDAP_ConfigurationList,
                                             p_NR_CellId,
                                             {v_GSM_MobilityInfo},
                                             v_SRBs);  // @sic R5-213442 sic@
            if (v_SRB == tsc_NR_RbId_SRB1) { // @sic R5s201654, R5s210672 sic@
              v_SRBs := SRB1_2;
              v_SRB := tsc_NR_RbId_SRB2;
            }
            if (p_DRBsAlreadyConfigured) {  // @sic R5-213442 sic@
              SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu))); // @sic R5s201413, R5s210672 sic@
              v_WaitForRRCComplete := true;  // @sic R5-215678 sic@
            } else {                        // @sic R5-213442 sic@
              // Send with a RRC Reconfiguration message
              f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                                 v_RadioBearerConfig,
                                                 v_RadioBearerList,
                                                 v_CellGroupConfig,
                                                 { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu)},
                                                 -, -, false); // @sic R5-215678 sic@
            }
            v_WaitForComplete := true;
            v_WaitForRRCComplete := true; // @sic R5-215678 sic@
            v_WaitForPDUSessionModificationComplete := true; //WA#6_2_2_2
            t_Wait.start;  // restart timer to wait for next Request
          }
        }
        else if (v_WaitForComplete and ischosen(v_GSMMsg.pdu_Session_Modification_Complete)) {
          // The v_GSM_MobilityInfo WILL be initialised when this iteration of the loop is processed
          v_WaitForPDUSessionModificationComplete := false; //WA#6_2_2_2
          if (not f_Check_NG_PDUSessionModificationComplete(v_NG_NAS_UL_Pdu, v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo.PTI)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION COMPLETE");
          }
          v_WaitForComplete := false;
          K := K + 1;
        }
        else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS PDU");
        }

      }
      v_RxdMsg := false; // @sic R5s210520 sic@
    }
    // Timer will have been started again in loop, but loop is now finished, so need to stop timer
    t_Wait.stop;
    
    return v_ReregisterIMS;
  }



